



De functie X.ZOEKEN()

In dit artikel gaat het over de relatief nieuwe functie X.ZOEKEN(). Met deze functie kun je in lijsten, tabellen en databases nagenoeg alles opzoeken. Van boven naar beneden, van links naar rechts, van rechts naar links en zelfs een reeks in een keer. Als je maar weet hoe!

In dit artikel leg ik uit wat je met X.ZOEKEN() kunt doen en waarom iedere Excel gebruiker deze functie gewoon zou moeten beheersen. Let op! Deze functie is alleen te gebruiken in Microsoft 365 en de webversie van Excel.

De functie uitgelegd

=X.ZOEKEN(**Zoekwaarde; Zoeken_matrix; Matrix_retourneren; Indien_niet_gevonden; Overeenkomstmodus; Zoekmodus**). De functie kent zes argumenten waarvan de eerste drie verplicht moeten worden ingevoerd.

Verplicht:

- Zoekwaarde: De waarde die moet worden gezocht (in een specifieke kolom van tabel)
- Zoeken-Matrix: De kolom, waarin de Zoekwaarde moet worden gezocht
- Matrix_retourneren: De kolom waaruit het resultaat moet worden opgehaald

De optionele opties komen in de loop van dit artikel aan bod.

Een voorbeeld

=X.ZOEKEN(B2;B5:B9;D5:D9) is een eenvoudige vorm van X.ZOEKEN().

	A	B	C	D
1		Netnr. Zoeken		
2		Mexico	+52	
3				
4		Land	Afk.	Netnr.
5		Brazilië	BR	+55
6		Rusland	RU	+7
7		Mexico	MX	+52
8		India	IN	+91
9		China	CN	+86

Een simpel bereik waaruit het netnummer van een land kan worden gezocht. Als eerste is de Zoekwaarde in B2 ingevoerd, daarna Zoeken-Matrix (B5:B9) en als laatste Matrix_retourneren (D5:D9). Met andere woorden: aan de hand van B2 wordt er gekeken of die waarde voorkomt in het bereik B5:B9, vervolgens wordt er gezocht in D5:D9 naar het resultaat.

Een uitgebreider voorbeeld

In feite komt het in dit voorbeeld op hetzelfde neer als het vorige voorbeeld. Er is alleen een optioneel argument aan toegevoegd. Het optionele argument betreft "indien_niet_gevonden". Deze optie geeft je de mogelijkheid een opmerking weer te geven indien de functie geen resultaat oplevert. Deze opmerking vervangt in dat geval de standaard waarde #N/B. Uiteraard hoef je dit argument niet te gebruiken om het resultaat te kunnen tonen.

```
=X.ZOEKEN(B4;$B$9:$B$17;$C$9:$C$17;"Niet gevonden")
```

```
X.ZOEKEN(zoekwaarde; zoeken-matrix; matrix_retourneren; [indien_niet_gevonden]; [overeenkomstmodus]; [zoekmodus])
```

De uitleg zoals bij het eerste voorbeeld is gegeven over de werking van de formule geldt ook voor deze formule. De formule in C4 is: =X.ZOEKEN(B4;B9:B17;C9:C17;"Niet gevonden")

Je typt de formule in C4 en drukt op Enter, dus gewoon zoals het behoort. Maar dan doet er zich een fenomeen voor, zonder dat je formules hebt getypt in de andere cellen verschijnt ook daar de uitkomst. In Excel noemt men dit **OVERLOOP** en komt tot uitdrukking in een blauwe rand om het bereik van de cellen die die resultaten bevatten. In dit geval is dat bereik B4:E4. Omdat de functie X.Zoeken verwijst naar een bereik dat vier kolommen betreft, in dit geval de kolommen B, C, D en E, weet de functie dat er resultaten moeten worden opgenomen in vier aaneengesloten cellen vanaf de cel waarin de functie is ingevoerd.

	A	B	C	D	E
1					
2					
3		Naam	Bedrag	Plaats	Medewerker
4		draai	€ 489,00	Barendrecht	vertegenw
5					
6		Zoekwaarde			
7		=X.ZOEKEN(B4;B9:B17;C9:C17;"Niet gevonden")			
8		Naam	bedrag	Plaats	Medewerker
9		Veldman	€ 712,00	Rotterdam	verkoper
10		de Vree	€ 998,25	den Haag	verkoper
11		Draai	€ 489,00	Barendrecht	vertegenw
12		de Vree	€ 1.022,00	Rotterdam	vertegenw
13		van Dalen	€ 898,98	Rotterdam	verkoper
14		de Goede	€ 1.099,45	Barendrecht	vertegenw
15		Online shop	€ 4.087,78	Magazijn	verkoper
16		van Dalen	€ 714,50	Schiedam	vertegenw
17		Groot	€ 1.022,33	Utrecht	verkoper
18					

Deze overloop cellen kun je niet wijzigen.

- Klik je een OVERLOOP-cel aan, dan wordt de tekst in de formulebalk licht grijs. Hiermee wordt aangegeven dat de functie of formule onderdeel is van een dynamische matrix formule
- Klik je vervolgens in de formulebalk om de inhoud te wijzigen, dan verdwijnt de functie en het resultaat in de betreffende cel
- Voer je een andere waarde in de cel in, dan verschijnt de melding #OVERLOOP!

3	Naam	Resultaat		
4	janse	Niet gevonden	Niet gevonden	Niet gevonden

Er is een niet bestaande naam ingevoerd en het resultaat is "niet gevonden", als reactie op de opgegeven waarde in "indien_niet_gevonden".

Net als de andere zoekfuncties in Excel vindt X.ZOEKEN() altijd de eerste waarde in het opgegeven bereik. De naam van de Vree staat tweemaal in het overzicht, dus zal de eerste waarde gevonden worden.

	A	B	C	D	E
19					
20	ID	Naam	Bedrag	Plaats	Medewerker
21	105	de Goede	€ 1.099,45	Barendrecht	vertegenw
22					
23	Zoekwaarde				
24		=X.ZOEKEN(B4;B9:B17;C9:C17;"Niet gevonden")			
25	ID	Naam	bedrag	Plaats	Medewerker
26	100	Veldman	€ 712,00	Rotterdam	verkoper
27	101	de Vree	€ 998,25	den Haag	verkoper
28	102	Draai	€ 489,00	Barendrecht	vertegenw
29	103	de Vree	€ 1.022,00	Rotterdam	vertegenw
30	104	van Dalen	€ 898,98	Rotterdam	verkoper
31	105	de Goede	€ 1.099,45	Barendrecht	vertegenw
32	106	Online shop	€ 4.087,78	Magazijn	verkoper
33	107	van Dalen	€ 714,50	Schiedam	vertegenw
34	108	Groot	€ 1.022,33	Utrecht	verkoper

Een idee zou kunnen zijn om elke rij een ID mee te geven. Het zoeken op een ID maakt het een stuk makkelijker om de juiste waarden te vinden. Dit overzicht bevat slechts enkele rijen met gegevens en is dus gemakkelijk te overzien. Maar ja, je zal een tabel of overzicht hebben met veel gegevens, dan is het natuurlijk een ander verhaal.

Genest zoeken

Met een geneste X.ZOEKEN() functie kun je zowel een verticale als een horizontale overeenkomst verkrijgen. De formule zoekt, in dit geval, eerst naar "Brutowinst" in B7:B11

en zoekt vervolgens naar het bijbehorende kwartaal in de bovenste rij van de tabel (C5:F5). Het retourneert ten slotte de waarde op het snijpunt van de twee. Dit is vergelijkbaar met het samen gebruiken van de functies [INDEX](#) en [VERGELIJKEN](#).

	A	B	C	D	E	F	G
1							
2		Kwartaal	Brutowinst	Nettowinst	Kosten	BTW	
3	Zoekwaarde	Kw1	25000	19342	25000	4246	
4			=X.ZOEKEN(C2;\$B7:\$B11;X.ZOEKEN(\$B3;\$C6:\$G6;\$C7:\$G11))				
5							
6		Winst/Verlie	Kw1	Kw2	Kw3	Kw4	
7		Verkoop	50.000	78.200	89.500	91.250	
8		Kosten	25.000	42.050	59.450	60.450	
9		Brutowinst	25.000	36.150	30.050	30.800	
10		BTW	4.246	6.211	5.346	5.298	
11		Nettowinst	19.342	28.295	24.352	24.134	
12							
13			Bedrag	Zoekresultaat			
14			4.246	BTW	=X.ZOEKEN(C13;C6:C10;B6:B10;0)		
15		Zoekwaarde					

De formule is in dit geval: =X.ZOEKEN(C2;\$B7:\$B11;X.ZOEKEN(\$B3;\$C6:\$G6;\$C7:\$G11)). Let op de dollartekens in de formule. Op de eerste zoekwaarde na (B2) zijn de kolommen dus absoluut, dat wil zeggen dat de kolommen "vaststaan". Deze formule moet doorgetrokken worden naar rechts. Hier bewijzen de dollartekens in de formule hun nut. In deze formulering gaat de optie OVERLOOP niet op.

Onderaan in de afbeelding zie je een voorbeeldje van zoeken van rechts naar links. De formule daarvoor is: =X.ZOEKEN(C14;C7:C11;B7:B11;0).

Zoeken met de overige niet optionele argumenten

We zoeken naar het bijbehorende percentage van een bepaald bedrag. De formule is als volgt: =X.ZOEKEN(B2;C7:C11;D7:D11;0;1;1). (afbeelding hieronder)

In kolom C wordt aan de hand van B2 gezocht naar het persoonlijke inkomen en vindt een overeenkomend belastingtarief in kolom D. Hier heeft het "indien_niet_gevonden" argument de waarde 0 (nul) en geeft dan ook 0 weer als er niets wordt gevonden. Het Overeenkomstmodus argument is ingesteld op 1, wat betekent dat de functie naar een exacte overeenkomst zal zoeken. Als deze niet kan worden gevonden, wordt de volgende grotere waarde geretourneerd. Ten slotte is in Zoekmodus argument ook 1 ingesteld, wat betekent dat met de functie wordt gezocht van het eerste item tot het laatste.

	A	B	C	D	E	F
1						
2		€ 450,00	22%	=X.ZOEKEN(B2;C7:C11;D7:D11;0;1;1)		
3						
4		22%	Draai	=X.ZOEKEN(B4;D7:D11;B7:B11;0;1;1)		
5						
6		Naam	bedrag	%		
7		Veldman	€ 712,00	9%		
8		de Vree	€ 998,25	12%		
9		Draai	€ 489,00	22%		
10		de Vree	€ 1.022,00	24%		
11		van Dalen	€ 898,98	32%		

In B2 is een bedrag ingevoerd dat niet in het overzicht staat. Dus wordt de dichtst bij zijnde grotere waarde teruggegeven. 22% is de uitkomst.

In C4 zie je nog een staaltje van rechts naar links zoeken. De formule is: =X.ZOEKEN(B4;D7:D11;B7:B11;0;1;1). Dus aan de hand van het aantal procenten wordt de naam gevonden.

De overeenkomst modus

- 0 - Exacte overeenkomst. Als niets wordt gevonden, wordt #N/A geretourneerd. Dit is de standaardinstelling.

- -1 - Exacte overeenkomst. Als niets wordt gevonden, wordt het volgende kleinere item geretourneerd.
- 1 - Exacte overeenkomst. Als niets wordt gevonden, wordt het volgende grotere item geretourneerd.

2 - Een overeenkomst met een jokerteken waar *,? en ~ die een [speciale betekenis](#) hebben.

De Zoekmodus kent vier waardes:

- 1 > Van boven naar beneden, vanaf het eerste item (Standaard)
- -1 > Van onder naar boven, vanaf het laatste item
- 2 > Binair, waarbij de Zoeken-Matrix in oplopende volgorde moet zijn gesorteerd. Als de Zoeken-Matrix niet is gesorteerd, worden er ongeldige resultaten geretourneerd
- -2 > Binair, waarbij de Zoeken-Matrix in aflopende volgorde moet zijn gesorteerd. Als de Zoeken-Matrix niet is gesorteerd, worden er ongeldige resultaten geretourneerd

Tot slot

- De functie X.Zoeken is een heel goede toevoeging op de al bestaande zoek functies.
- Het kunnen zoeken in een kolom ongeacht op welke positie die kolom in de tabel staat. De functie X.Zoeken heeft een groot voordeel ten opzichte van de functie Vert.Zoeken
- De mogelijkheid om een opmerking of waarde te tonen indien een zoekopdracht niet tot een overeenkomst leidt is een praktisch voordeel dat veel functies niet hebben
- Het kunnen ophalen van resultaten uit meerdere kolommen door de functie in één cel in te voeren.

Een voorbeeld van X.VERGELIJKEN()

Met de functie **X.VERGELIJKEN()** wordt gezocht naar een opgegeven item in een cellenbereik en vervolgens wordt de relatieve positie van het item geretourneerd.

	A	B	C	D	E	F	G
1							
2		Naam	Maand	Bedrag			
3		draai	mrt	583,5			
4		=INDEX(C7:E15;X.VERGELIJKEN(B3;B7:B15);X.VERGELIJKEN(C3;C6:E6))					
6		Naam	jan	feb	mrt		
7		Veldman	€ 712,00	€ 1.068,00	€ 918,00		
8		de Vree	€ 998,25	€ 1.497,38	€ 1.347,38		
9		Draai	€ 489,00	€ 733,50	€ 583,50		
10		de Vree	€ 1.022,00	€ 1.533,00	€ 1.383,00		
11		van Dalen	€ 898,98	€ 1.348,47	€ 1.198,47		
12		de Goede	€ 1.099,45	€ 1.649,18	€ 1.499,18		
13		Online shop	€ 4.087,78	€ 6.131,67	€ 5.981,67		
14		van Dalen	€ 714,50	€ 1.071,75	€ 921,75		
15		Groot	€ 1.022,33	€ 1.533,50	€ 1.383,50		

De functie VERGELIJKEN() zoekt een item en geeft daarvan de positie weer. De functie X.VERGELIJKEN() doet hetzelfde, maar met minder argumenten. Met de functies INDEX() en VERGELIJKEN() kun je hetzelfde resultaat behalen als in het voorbeeld hierboven. Alleen zijn daar meer argumenten voor nodig dan met de functies INDEX() samen met X.VERGELIJKEN(). Voer in B3 een naam in en in C3 de maand. Met de formule =INDEX(C7:E15;X.VERGELIJKEN(B3;B7:B15);X.VERGELIJKEN(C3;C6:E6)) in D3 wordt de juiste uitkomst weergegeven.

In de functie X.VERGELIJKEN() kun je wel de Overeenkomstmodus en Zoekmodus opgeven, maar "indien_niet_gevonden" niet. Gaat er iets fout met de formule dan moet je het met de foutwaarde #N/B doen.